

# Oefeningen

## Video: Zoeken en presenteren van SNOMED-concepten in Snowstorm

Usecase 1: Zoeken op tekst en het zoekresultaat tonen

*Scenario: Een arts heeft een PTA (percutane transluminale angioplastiek) uitgevoerd en voert deze zoekterm in. Welk concept, met welke voorkeursterm, legt ze vast in het EPD?*

1. Zoek op PTA en noteer zowel het conceptId van het beste zoekresultaat, als de beschrijving die gematcht heeft op de zoekterm.
2. Haal alle beschrijvingen op van het betreffende conceptId.
3. Identificeer de medische voorkeursterm. Presenteer zowel deze als de beschrijving die gematcht heeft aan de arts, zodat zij kan controleren of dat inderdaad het concept is dat ze bedoelt.

Usecase 2: Concept presenteren in het PGO

*Scenario: Het in usecase 1 vastgelegde concept wordt doorgestuurd naar het PGO van de patiënt. Met welke beschrijving wordt het daar getoond in het PGO?*

4. Identificeer de patiëntvriendelijke voorkeursterm. Hiermee wordt het concept getoond in het PGO.

Voer nogmaals usecase 1 & 2 uit met bv. pneumonie, cystische fibrose, trisomie-21, of een zoekterm naar keuze.

Usecase 3: Concept met ontbrekende beschrijvingen presenteren

*Aanvullend scenario: Een arts legt een concept vast dat in SNOMED geen patiëntvriendelijke term heeft (bv. 54556006 |fractuur van ulna (aandoening)|), of in het geheel niet vertaald is (bv. 90332006 |Product containing paracetamol (medicinal product)|). Wat toont het PGO?*

5. Vraag alle beschrijvingen van het zoekresultaat. Toon de eerste term die je vindt:
  - a. Patiëntvriendelijke Nederlandse voorkeursterm.
  - b. Nederlandse voorkeursterm voor zorgverleners.
  - c. Amerikaanse voorkeursterm.

## Video: Queryen in SNOMED met Snowstorm

### Usecase 1: De hiërarchie verkennen

*Scenario: Een arts heeft een MRI van het hart uitgevoerd met het contrastmiddel dobutamine. Ze wil dit vastleggen. Omdat ze niet zeker weet hoe gedetailleerd de lijst van verrichtingen is, zoekt ze op 'MRI hart' en verkent de hiërarchische afstammelingen. Welk concept legt ze uiteindelijk vast?*

1. Zoek de beste match voor 'MRI hart'.
2. Haal de kinderen op van dit concept. Deze kun je tonen in een boomstructuur.
3. Haal van elk concept dat jij als arts zou willen openklappen ook de kinderen op, totdat je het juiste concept gevonden hebt.

### Usecase 2: De onderzoeker

*Scenario: Een onderzoeker wil analyseren hoe vaak infecties met Streptokokken groep A voorkomen in verschillende ziekenhuizen. Alle ziekenhuizen gebruiken SNOMED, maar wel verschillende concepten uit SNOMED. Met een ECL-query zoekt de onderzoeker in SNOMED alle aandoeningen veroorzaakt door Streptokokken groep A. De juiste ECL-query is:*

```
<< 64572001 |Disease (disorder)|:  
    246075003 |Causative agent (attribute)| = << 80166006  
    |Streptococcus pyogenes (organism)|
```

4. Hoeveel verschillende SNOMED-concepten matchen de query?
5. Hoeveel vormen van sepsis bevat SNOMED?
6. Hoeveel verrichtingen op de humeruskop bevat SNOMED?  
Tip: gebruik << 363704007 |Procedure site| als attribuut).

## Appendix A: Links en Id's

Snowstorm: <https://snowstorm.test-nictiz.nl/swagger-ui.html>

Art-Decor waardenlijstindex: <http://decor.nictiz.nl/decor/services/ProjectIndex?prefix=zib2017bbr->

ZIBs: <https://zibs.nl/wiki/Hoofdpagina>


Nictiz-browser: <http://terminologie.nictiz.nl/art-decor/snomed-ct>

Internationale browser: <http://browser.ihtsdotools.org/>

### Referentiesets:

- |   |                    |
|---|--------------------|
| • Medisch-Nederlandse taalreferentieset | 31000146106        |
| • Patiëntvriendelijke taalreferentieset | 15551000146102     |
| • Amerikaanse taalreferentieset         | 900000000000509007 |
| • Britse taalreferentieset              | 900000000000508004 |
| • Landelijk implantatenregister (LIR)   | 52801000146101     |
| • GDMN-mapping                          | 467614008          |
| • Micro-organismen                      | 2581000146104      |
| • ICD-10-mapping                        | 447562003          |

## Appendix B: Expression constraint language



Figuur 1: Componenten van een expressiebepanking

Figuur 1 toont een voorbeeld waarin de verschillende onderdelen gemarkeerd zijn. In de volgende paragrafen worden de onderdelen toegelicht aan de hand van voorbeelden.

### Constraint operator

De eenvoudigste expressiebepanking is een precoördinatie:

```
404684003 | clinical finding |
```

Deze expressie is equivalent aan dezelfde expressie in de compositionele grammatica en definieert een set met één element.

Een expressiebepanking kan voorafgegaan worden door een *constraint operator*, die in korte of volledige syntax kan worden weergegeven (zie Tabel 1). Zo zijn de volgende expressies equivalent en definiëren allebei de set van afstammelingen van *clinical finding*, *exclusief* de *clinical finding* zelf:

```
<404684003 | clinical finding |
```

```
descendantOf 404684003 | clinical finding |
```

De volledige syntax moet gevolgd worden door een spatie, bij de korte syntax is de spatie optioneel.

Operators	Korte syntax	Volledige syntax (case insensitive, gevolgd door spatie)
Afstammeling	<	Descendantof
Afstammeling of zelf	<<	Descendantorselfof
Voorouder	>	Ancestorof
Voorouder of zelf	>>	Ancestororselfof
Lid van	^	Memberof
Any	*	Any

Tabel 1: Constraint operators

De set van afstammelingen *inclusief* *clinical finding* kan worden gedefinieerd met de volgende expressie:

```
<< 404684003 | clinical finding |
```

De operators om voorouders aan te duiden worden op vergelijkbare wijze toegepast. Constraint operators kunnen op concepten maar ook op expressies worden toegepast.

### Refinement

Verfijning gaat grotendeels op dezelfde wijze als in de compositionele grammatica (zie paragraaf 1.4). Zo wordt de set van longaandoeningen met morfologie 'oedeem' als volgt gedefinieerd:

```
< 19829001 |disorder of lung|:
  116676008 |associated morphology| = 79654002 |edema|
```

De waarde van het attribuut kan verfijnd worden met een constraint operator, zodat we de set van longaandoeningen met een *soort* oedeem als volgt kunnen definiëren:

```
< 19829001 |disorder of lung|:
  116676008 |associated morphology| = << 79654002 |edema|
```

Ook het attribuut kan een constraint operator krijgen. Zo vallen in het Content Model de eigenschappen *After*, *Due to* en *Causative agent* allemaal onder *Associated with*. De set van klinische bevindingen die een soort associatie hebben met een soort oedeem wordt omschreven door:

```
<< 404684003 |clinical finding|:
  << 47429007 |associated with| = << 267038008 |edema|
```

Voor attribuutgroepen en genestelde verfijningen, zie paragrafen 1.5 en 1.6.

### Concrete waarden

Het Content Model bevat sinds kort attributen die concrete waarden kunnen hebben. Met de Expression Constraint syntax kunnen aan die waarden beperkingen opgelegd worden. De volgende expressie beschrijft de set van medicijnen met hooguit twee actieve ingrediënten:

```
373873005 |pharmaceutical / biologic product|:
  100005 |active ingredient count| <= #2
```

Operator	Functie	Gebruikt voor:
=	Gelijk aan	Getallen en tekst
!=	Ongelijk aan	Getallen en tekst
<>	Ongelijk aan	Getallen en tekst
not =	Ongelijk aan	Getallen en tekst
>	Groter dan	Getallen
>=	Groter dan of gelijk aan	Getallen
<	Kleiner dan	Getallen
<=	Kleiner dan of gelijk aan	Getallen

Tabel 2: Operatoren om waarden te vergelijken.

De set van amoxicillinetabletten met een sterkte tussen de 500 en 800 milligram:

```
< 111115 |amoxicillin tablet|:
  { 127489000 |has active ingredient| = 372687004 |amoxicillin|,
    111115 |has basis of strength| = ( 372687004 |amoxicillin|,
      111115 |strength magnitude| >= #500, 111115 |strength magnitude| <= #800,
      111115 |strength unit| = 258684004 |mg|)}
```

## Reversed attributes

Meestal (en in alle bovenstaande gevallen) wordt een verfijning gedefinieerd door aan een bronconcept een attribuut toe te voegen met een beperking op de waarde van dat attribuut. Soms wil je de beperking echter andersom opleggen, omdat je de set zoekt van concepten die niet een bepaald attribuut hebben, maar die de waarde van een attribuut zijn. Om dit te bereiken kunnen we het attribuut reverseren oftewel omdraaien met een hoofdletter R of de string *reverseOf*. De volgende expressie beschrijft de set van substanties die de actieve ingrediënten zijn van een Triphasiltablet:

```
< 105590001 |substance|:  
  R 127489000 |has active ingredient| = 111115 |TRIPHASIL tablet|
```

## Conjunctie en disjunctie

Bij default moet aan alle voorwaarden van een expressiebeperking worden voldaan. Dit noemen we *conjunctie*. De volgende expressies zijn dus equivalent:

```
< 404684003 |clinical finding|:  
  363698007 |finding site| = << 39057004 |pulmonary valve structure|,  
  116676008 |associated morphology| = << 415582006 |stenosis|  
  
< 404684003 |clinical finding|:  
  363698007 |finding site| = << 39057004 |pulmonary valve structure| AND  
  116676008 |associated morphology| = << 415582006 |stenosis|
```

Met *disjunctie* kun je een 'of ... of'-beperking specificeren. Zo kun je met de volgende expressie alle aandoeningen vinden die veroorzaakt worden door een projectiel of door vibratie:

```
< 404684003 |clinical finding|:  
  42752001 |due to| = << 44468009 |missile| OR  
  42752001 |due to| = << 33679000 |vibration|
```

Conjunctie en disjunctie kunnen ook binnen een attribuutwaarde gebruikt worden. De volgende beperking is dus equivalent met de vorige:

```
< 404684003 |clinical finding|:  
  42752001 |due to| = (<< 44468009 |missile| OR << 33679000 |vibration|)
```

Je kunt conjunctie en disjunctie meerdere keren gebruiken in een expressie en ook door elkaar. Zorg daarbij wel dat de betekenis onambigu is door waar nodig ronde haken te plaatsen! De volgende expressies hebben namelijk verschillende betekenissen:

```
< 404684003 |clinical finding| :  
  ( 363698007 |finding site| = << 39057004 |pulmonary valve structure| AND  
  116676008 |associated morphology| = << 415582006 |stenosis| ) OR  
  42752001 |due to| = << 445238008 |malignant carcinoid tumor|  
  
< 404684003 |clinical finding| :  
  363698007 |finding site| = << 39057004 |pulmonary valve structure| AND  
  ( 116676008 |associated morphology| = << 415582006 |stenosis| OR  
  42752001 |due to| = << 445238008 |malignant carcinoid tumor| )
```

De eerste matcht alle concepten die òf zowel het *finding site*- als het *associated morphology*-attribuut hebben, òf het *due to* attribuut; de tweede matcht alle concepten die zowel het *due to*-attribuut als òf de *finding site*, òf de *associated morphology* hebben. De ronde haken bepalen de betekenis: zonder haakjes zou de expressie invalide zijn.

De disjunctie is overigens inclusief. Dat wil zeggen dat een expressie voldoet aan de disjunctie als het voldoet aan één van de voorwaarden maar ook als het aan allebei de voorwaarden voldoet. Een concept dat alledrie genoemde attributen bezit, voldoet dus aan beide restricties!

### Meerdere focusconcepten

Met conjunctie en disjunctie kun je ook meerdere focusconcepten opgeven. De set van alle concepten die zowel een longaandoening als een rompoedeem zijn wordt als volgt gedefinieerd:

```
< 19829001 |disorder of lung| AND < 301867009 |edema of trunk|
```

Met disjunctie kunnen we de set van alle longaandoeningen en rompoedemen beschrijven, oftewel alle concepten die onder één of beide concepten vallen:

```
< 19829001 |disorder of lung| OR < 301867009 |edema of trunk|
```

Ook hier geldt dat de disjunctie inclusief is. Alle concepten die voldoen aan de eerste expressie (zowel longaandoening als rompoedeem) voldoen dus automatisch ook aan de tweede; de eerste expressie beschrijft een subset van de tweede.

## 2.9 Exclusie

Met exclusie kun je uitzonderingen definiëren. Om bijvoorbeeld alle longaandoeningen op te halen die niet onder rompoedeem vallen, gebruik je de volgende beperking:

```
< 19829001 |disorder of lung| MINUS < 301867009 |edema of trunk|
```

Je kunt exclusie ook gebruiken om alle leden van een referentieset uit te sluiten:

```
< 19829001 |disorder of lung| MINUS ^ 152725851000154106 |cardiology reference set|
```

En ook binnen een attribuutwaarde:

```
< 404684003 |clinical finding|: 116676008 |associated morphology| =  
  (( < 56208002 |ulcer| AND < 50960005 |hemorrhage|) MINUS < 26036001  
  |obstruction|)
```

Een andere manier om exclusies te definiëren is met de *not equals* operator (geschreven als '!=' of 'NOT ='). Hiermee kun je bijvoorbeeld aangeven dat je alle bevindingen wilt hebben met een morfologie die niet een soort oppervlakkige verwonding is:

```
< 404684003 |clinical finding|:  
  116676008 |associated morphology| != < 3380003 |Superficial injury|
```

Voor meer informatie over de syntax en mogelijke toepassingen ervan, zie het document waarop deze samenvatting gebaseerd is: de [SNOMED CT Expression Constraint Specification and Guide](#).

## Appendix C: Antwoorden

### Video: Zoeken en presenteren van SNOMED-concepten

1. Het juiste concept is 68457009 | percutane transluminale ballonangioplastiek (verrichting) |. Het eerste zoekresultaat is weliswaar 46572007 | stollingsfactor XI (substantie) |, dat ook een synoniem PTA heeft, maar dat verwijst naar iets heel anders... Om dergelijke verwarring te voorkomen, raden wij ten strengste aan om een concept altijd te tonen met de voorkeursterm en niet met het synoniem.
2. In de September 2020-editie zou u 12 beschrijvingen moeten vinden.
3. De medische voorkeursterm is *percutane transluminale ballonangioplastiek*.
4. De patiëntvriendelijke voorkeursterm is *dotteren*.

### Video: Queryen in SNOMED met Snowstorm

1. De beste match voor 'MRI hart' is natuurlijk 241620005 |MRI van hart|.
2. Dit concept heeft 14 kinderen, waaronder 736793003 |MRI van hart met contrast|.
3. De beste match is het concept 95591000146105 |MRI van hart met contrast onder toediening van dobutamine|.
4. In de September 2020-editie voldoen 24 concepten aan deze query.
5. De meest eenvoudige correcte query is `<< 91302008 |sepsis|`. Dit levert in de September 2020-editie 117 resultaten op.
6. De juiste query is  
`<< 71388002 |verrichting|:`  
`<< 363704007 |locatie van verrichting| = << 81965001 |structuur van caput humeri|`  
Dit levert in de September 2020-editie 22 resultaten op.